// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/BatchVote.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.19;

import { Ownable } from "@openzeppelin/contracts/access/Ownable.sol";
import { Pausable } from "@openzeppelin/contracts/security/Pausable.sol";
import { IAddressContract } from "./interfaces/IAddressContract.sol";


// barac contract interface used at the time of delegaion.
interface IBarac {
    function delegateBySig(
        address delegatee,
        uint256 nonce,
        uint256 expiry,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;
}

// Dao contract interface used at the time of voting.
interface IDao {
    function castVoteBySig(
        uint256 proposalId,
        bool support,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    function castVoteJudgementBySig(
        uint256 proposalId,
        bool support,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;
}

contract BatchVote is Ownable, Pausable
{
    // Dao contract address
    address public dao;

    // barac contract address
    address public barac;

    // Structure used for batch delegate
    struct DelegateSignature {
        address delegatee;
        uint256 nonce;
        uint256 expiry;
        uint8 v;
        bytes32 r;
        bytes32 s;
    }

    // Structure used for batch vote
    struct CastVoteSignature {
        uint256 proposalId;
        bool voteFor; 
        uint8 v; 
        bytes32 r; 
        bytes32 s;
    }

  
    function setContractAddresses(IAddressContract _contractFactory) external onlyOwner {
        dao = _contractFactory.getDao();
        barac = _contractFactory.getBarac();
    }


    /**
     * @notice Pause batch vote contract.
     * @dev Owner can pause the contract.
     */
    function pause() external onlyOwner {
        _pause();
    }

    /**
     * @notice Unpause batch vote contract.
     * @dev Owner can un-pause the contract.
     */
    function unpause() external onlyOwner {
        _unpause();
    }

    /**
     * @notice Batch delegate using user signatures.
     * @dev Any user can call this function to delegate the user addresses.
     */
    function delegateBySigs(DelegateSignature[] memory sigs)
        external
        whenNotPaused
    {
        for (uint256 i = 0; i < sigs.length; i++) {
            DelegateSignature memory sig = sigs[i];
            IBarac(barac).delegateBySig(
                sig.delegatee,
                sig.nonce,
                sig.expiry,
                sig.v,
                sig.r,
                sig.s
            );
        }
    }

    /**
     * @notice Batch voting using user signatures.
     * @dev Any user can call this function to cast votes in batch.
     */
    function castVoteBySigs(CastVoteSignature[] memory sigs)
        external
        whenNotPaused
    {
        for (uint256 i = 0; i < sigs.length; i++) {
            CastVoteSignature memory sig = sigs[i];
            IDao(dao).castVoteBySig(
                sig.proposalId,
                sig.voteFor,
                sig.v,
                sig.r,
                sig.s
            );
        }
    }

    /**
     * @notice Batch voting using user signatures.
     * @dev Any user can call this function to cast votes in batch.
     */
    function castVoteJudgementBySigs(CastVoteSignature[] memory sigs)
        external
        whenNotPaused
    {
        for (uint256 i = 0; i < sigs.length; i++) {
            CastVoteSignature memory sig = sigs[i];
            IDao(dao).castVoteJudgementBySig(
                sig.proposalId,
                sig.voteFor,
                sig.v,
                sig.r,
                sig.s
            );
        }
    }
}
// File: contracts/interfaces/IAddressContract.sol



//SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;


interface IAddressContract {

    function getDao() external view returns (address);
    
    function getTreasury() external view returns (address);
   
    function getScarabNFT() external view returns (address);
    
    function getScarab() external view returns (address);

    function getBarac() external view returns (address);
    
}